Euler Angles

Hendrik van Hees

March 31, 2020

Contents

1 Rotation matrices

Rotation matrices are most easily introduced as transformation matrices between right-handed Cartesian bases

$$\vec{e}'_k = \vec{e}_j D^j{}_k. \tag{1}$$

Here and in the following we use the Einstein summation convention, i.e., over pairs of repeated indices one has to sum. In this formula we have to sum over $j \in \{1,2,3\}$.

The properties of the rotation matrices follows from the definition of right-handed Cartesian bases. First such a basis builds an orthonormal system, i.e.,

$$\vec{e}'_k \cdot \vec{e}'_l = \delta_{kl} \Rightarrow \vec{e}_j \cdot \vec{e}_m D^j{}_k D^m{}_l = \delta_{jm} D^j{}_k D^m{}_l = D^j{}_k D^j{}_l = \delta_{kl}, \tag{2}$$

from which

$$(D^{-1})^l{}_j = D^j{}_l. \tag{3}$$

In matrix notation that means

$$\hat{D}^{-1} = \hat{D}^{\mathrm{T}}, \tag{4}$$

where the T denotes transposition, i.e., to write the columns of the matrix $\hat{D}$ as the rows of $\hat{D}^{\mathrm{T}}$.

Finally the right-handedness of both bases means, using the Levi-Civita symbol, ϵ_{jkl},

$$\vec{e}'_j \cdot (\vec{e}'_k \times \vec{e}'_l) = \epsilon_{jkl} = \vec{e}'_m \cdot (\vec{e}_n \times \vec{e}_o) D^m{}_j D^n{}_k D^o{}_l = \epsilon_{mno} D^m{}_j D^n{}_k D^o{}_l = \det \hat{D} \epsilon_{jkl} \Rightarrow \det \hat{D} = +1. \tag{5}$$

Rotation matrices are orthonormal, i.e., $\hat{D}\hat{D}^{\mathrm{T}} = \hat{D}^{\mathrm{T}}\hat{D} = \mathbb{1}$ and obey $\det \hat{D} = 1$. These matrices build a group since it is easy to show that with two such matrices $\hat{D}_1$ and $\hat{D}_2$ also $\hat{D}_1\hat{D}_2$ also fulfill the conditions. This group is called the special orthogonal group in 3D Euclidean vector spaces, SO(3).



Figure 1: Definition of the Euler angles.

2 Euler angles

Euler angles are a special parametrization of all SO(3) matrices. Given two Cartesian righthanded bases $\vec{e}_j$ and $\vec{e}_k'$, as shown in Fig. 1 we transfer first the vector $\vec{e}_1$ by a rotation around the axis $\vec{e}_3$ by the angle ψ into the direction of the intersection line between the plane spanned by $\vec{e}_1$ and $\vec{e}_2$ and the plane spanned by $\vec{e}_1'$ and $\vec{e}_2'$, given by the unit vector $\vec{k}$, which is the $\vec{e}_1''$ axis of an intermediate Cartesian basis $\vec{e}_k''$. Of course $\vec{e}_3'' = \vec{e}_3$. The next rotation is around this axis the the angle ϑ such that $\vec{e}_3$ is rotated into $\vec{e}_3'$ and finally a third rotation around $\vec{e}_3'$ by the angle φ such as to rotate the line of nodes, $\vec{k} = \vec{e}_1''$ into $\vec{e}_1'$. The resulting rotation matrix is then given by

$$\begin{aligned} \hat{D} &= \hat{D}^{(3)}(\psi)\hat{D}^{(1)}(\vartheta)\hat{D}^{(3)}(\phi) \\ &= \begin{pmatrix} \cos\psi\cos\varphi - \sin\psi\cos\vartheta\sin\varphi & -\cos\psi\sin\phi - \sin\psi\cos\vartheta\cos\varphi & \sin\psi\sin\vartheta \\ \sin\psi\cos\varphi + \cos\psi\cos\vartheta\sin\varphi & -\sin\psi\sin\varphi + \cos\psi\cos\vartheta\cos\varphi & -\cos\psi\sin\vartheta \\ \sin\vartheta\sin\varphi & \sin\vartheta\cos\varphi & \cos\vartheta \end{pmatrix} \end{aligned} \tag{6}$$

such that (1) holds. From the above intuitive description it's clear that $\psi, \varphi \in]-\pi, \pi]$ and $\vartheta \in [0, \pi]$ covers the entire SO(3) once.

Here

$$\hat{D}^{(1)}(\vartheta) = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \cos\vartheta & -\sin\vartheta \\ 0 & \sin\vartheta & \cos\vartheta \end{pmatrix}, \quad \hat{D}^{(3)}(\psi) = \begin{pmatrix} \cos\psi & -\sin\psi & 0 \\ \sin\psi & \cos\psi & 0 \\ 0 & 0 & 1 \end{pmatrix} \tag{7}$$

denote rotations around the $\vec{e}_1$ or $\vec{e}_3$ axis, respectively.

Now we formally prove this algebraically. To that end assume that $\hat{O}=(O^j{}_k)\in \mathrm{SO}(3)$. We first assume that $O^3{}_3 \notin \{-1,1\}$. Then the angle ϑ is determined uniquely by

$$\cos\vartheta = O^3{}_3 \Rightarrow \vartheta = \arccos(O^3{}_3) \tag{8}$$

and $\vartheta \in]0,\pi[$. Since $\hat{O}$ is a orthonormal matrix this implies[1]

$$(O^1{}_3)^2 + (O^2{}_3)^2 = 1-(O^3{}_3)^2 = 1-\cos^2\vartheta = \sin^2\vartheta, \tag{9}$$

and thus we can determine the angle $\psi \in]-\pi,\pi]$ uniquely by defining

$$\sin\psi = \frac{1}{\sin\vartheta} O^1{}_3, \quad \cos\psi = -\frac{1}{\sin\vartheta} O^2{}_3 \tag{10}$$

leading to

$$\psi = \operatorname{sign} O^1{}_3 \arccos\left(-\frac{1}{\sin\vartheta} O^2{}_3\right) \in]-\pi,\pi]. \tag{11}$$

By the same argument one finds

$$\sin\varphi = \frac{1}{\sin\vartheta} O^3{}_1, \quad \cos\varphi = \frac{1}{\sin\vartheta} O^3{}_2 \tag{12}$$

and thus

$$\varphi = \operatorname{sign} O^3{}_1 \arccos\left(\frac{1}{\sin\vartheta} O^3{}_2\right) \in]-\pi,\pi]. \tag{13}$$

So far we have chosen the three angles such that

$$\hat{O} = \begin{pmatrix} O^1{}_1 & O^1{}_2 & \sin\psi\sin\vartheta \\ O^2{}_1 & O^2{}_2 & -\cos\psi\sin\vartheta \\ \sin\vartheta\sin\varphi & \sin\vartheta\cos\varphi & \cos\vartheta \end{pmatrix}. \tag{14}$$

Now we evaluate

$$\hat{B} = \hat{D}^{-1}\hat{O} = \hat{D}^{\mathrm{T}}\hat{O}. \tag{15}$$

The matrix elements are pretty complicated combinations of trigonometric terms with the four not yet parametrized matrix elements of (10)[2]. By construction, however the 3rd column of $\hat{B}$ is $(B^1{}_3, B^2{}_3, B^3{}_3)^{\mathrm{T}} = (0,0,1)^{\mathrm{T}}$. Now $\hat{B}$ is an SO(3) matrix and thus its columns are three orthonormal vectors in $\mathbb{R}^3$. This implies for the 3rd row

$$B^3{}_1 = B^3{}_2 = 0. \tag{16}$$

This implies that

$$\hat{B} = \begin{pmatrix} B^1{}_1 & B^1{}_3 & 0 \\ B^2{}_1 & B^2{}_2 & 0 \\ 0 & 0 & 1 \end{pmatrix}. \tag{17}$$

Since $\hat{B} \in \mathrm{SO}(3)$ this implies that

$$B^1{}_1 = B^2{}_2, \quad B^1{}_2 = -B^2{}_1. \tag{18}$$

[1]Note that $\sin\vartheta \neq 0$ here!

[2]See the Appendix with a Mathematica notebook with the details of the matrix-manipulation results.

Using $\det \hat{B} = 1$, Eq. (16) and the 1st equation (18) finally leads to $\hat{B} = \mathbb{1}$ and thus $\hat{O} = \hat{D}\hat{B} = \hat{D}$, i.e., we have indeed determined three angles ψ, ϑ, and φ, such that $\hat{D} = \hat{O}$[3].

So far we have excluded the cases ${O^3}_3 = 1$ and ${O^3}_3 = -1$. In the first case from (6) we have $\vartheta = 0$ and then $\hat{D} = \hat{D}_3(\psi)\hat{D}_3(\phi) = \hat{D}_3(\psi+\phi)$. Of course if ${O^3}_3 = 1$ we have a rotation around the $\vec{e}_3$ axis, and we have just to choose $\psi + \varphi$ as the corresponding angle, i.e., we can still parametrize the rotation matrix $\hat{O}$ with Euler angles with $\vartheta = 0$, but only $\psi + \varphi$ is determined, but ψ and φ are indetermined. This implies that the parametrization of the SO(3) with Euler angles is degenerate, i.e., one has a "coordinate singularity" for rotations around the $\vec{e}_3$-axis though of course there's nothing special of such rotations in SO(3) as a differentiable manifold (actually a Lie group). This is also known as the "gimbal lock", because it describes a singular configuration of the well-known gimbals used to realize free tops in the mechanics lecture[4].

Finally we consider the case ${O^3}_3 = -1$, which through (6) leads to $\vartheta = \pi$. By direct calculation we get

$$\hat{D} = \hat{D}_3(\psi)\hat{D}_1(\pi)\hat{D}_3(\varphi) = \begin{pmatrix} \cos(\varphi-\psi) & -\sin(\varphi-\psi) & 0 \\ -\sin(\varphi-\psi) & -\cos(\varphi-\psi) & 0 \\ 0 & 0 & -1 \end{pmatrix}. \tag{19}$$

Since $\hat{O}$ is an SO(3) matrix with ${O^3}_3 = -1$, we can determine $\varphi - \psi \in [0, 2\pi[$ uniquely, but φ and ψ are again undetermined, and we have just another "gimbal lock".

A Mathematica notebook for the proof

Here we reproduce the Mathematica notebook with the matrix manipulations necessary for the above given proof for the parameterizability of an arbitrary matrix $\hat{O} \in \mathrm{SO}(3)$ by Euler angles.

[3]For details of the calculation with help of Mathematica, see the Appendix.

[4]See the Wikipedia page with nice animations for gimbals: `https://en.wikipedia.org/wiki/Gimbal_lock`.

```
In[1]:= D3[psi_] = {{Cos[psi], -Sin[psi], 0}, {Sin[psi], Cos[psi], 0}, {0, 0, 1}}
Out[1]= {{Cos[psi], -Sin[psi], 0}, {Sin[psi], Cos[psi], 0}, {0, 0, 1}}

In[2]:= D1[th_] = {{1, 0, 0}, {0, Cos[th], -Sin[th]}, {0, Sin[th], Cos[th]}};

In[3]:= ompr = {thdot Cos[ph] + psdot Sin[th] Sin[ph], -thdot Sin[ph] + psdot Sin[th] Cos[ph], phdot + psdot Cos[th]};

In[4]:= Dmat = D3[psi].D1[th].D3[ph];

In[5]:= MatrixForm[Dmat]
Out[5]//MatrixForm=
( Cos[ph] Cos[psi] - Cos[th] Sin[ph] Sin[psi]   -Cos[psi] Sin[ph] - Cos[ph] Cos[th] Sin[psi]   Sin[psi] Sin[th]
  Cos[psi] Cos[th] Sin[ph] + Cos[ph] Sin[psi]    Cos[ph] Cos[psi] Cos[th] - Sin[ph] Sin[psi]   -Cos[psi] Sin[th]
  Sin[ph] Sin[th]                                Cos[ph] Sin[th]                                Cos[th] )

In[6]:= Omat = {{O11, O12, Sin[psi] Sin[th]}, {O21, O22, -Cos[psi] Sin[th]}, {Sin[ph] Sin[th], Cos[ph] Sin[th], Cos[th]}};

In[7]:= MatrixForm[Omat]
Out[7]//MatrixForm=
( O11               O12               Sin[psi] Sin[th]
  O21               O22               -Cos[psi] Sin[th]
  Sin[ph] Sin[th]   Cos[ph] Sin[th]   Cos[th] )

In[8]:= B = FullSimplify[Transpose[Dmat].Omat]
Out[8]= {{Cos[ph] (O11 Cos[psi] + O21 Sin[psi]) + Sin[ph] (O21 Cos[psi] Cos[th] - O11 Cos[th] Sin[psi] + Sin[ph] Sin[th]^2),
    Cos[th] Sin[ph] (O22 Cos[psi] - O12 Sin[psi]) + Cos[ph] (O12 Cos[psi] + O22 Sin[psi] + Sin[ph] Sin[th]^2), 0},
  {-Sin[ph] (O11 Cos[psi] + O21 Sin[psi]) + Cos[ph] (O21 Cos[psi] Cos[th] - O11 Cos[th] Sin[psi] + Sin[ph] Sin[th]^2),
    Cos[ph] Cos[th] (O22 Cos[psi] - O12 Sin[psi]) - Sin[ph] (O12 Cos[psi] + O22 Sin[psi]) + Cos[ph]^2 Sin[th]^2, 0},
  {(-O21 Cos[psi] + Cos[th] Sin[ph] + O11 Sin[psi]) Sin[th], (-O22 Cos[psi] + Cos[ph] Cos[th] + O12 Sin[psi]) Sin[th], 1}}

In[9]:= erg1 = FullSimplify[Solve[{B[[3]][[1]] == 0, B[[3]][[2]] == 0}, {O11, O12}]]
Out[9]= {{O11 → O21 Cot[psi] - Cos[th] Csc[psi] Sin[ph], O12 → O22 Cot[psi] - Cos[ph] Cos[th] Csc[psi]}}
```

```
In[10]:= B2 = FullSimplify[B /. Flatten[erg1]]

Out[10]= {{Sin[ph]^2 + Cos[ph] (O21 Cos[psi] Cot[psi] - Cos[th] Cot[psi] Sin[ph] + O21 Sin[psi]),
     Cos[ph] (O22 Cos[psi] Cot[psi] - Cos[ph] Cos[th] Cot[psi] + Sin[ph] + O22 Sin[psi]), 0},
    {Sin[ph] (Cos[ph] - O21 Cos[psi] Cot[psi] + Cos[th] Cot[psi] Sin[ph] - O21 Sin[psi]),
     Cos[ph]^2 + Cos[ph] Cos[th] Cot[psi] Sin[ph] - O22 Csc[psi] Sin[ph], 0}, {0, 0, 1}}

In[11]:= erg2 = Solve[FullSimplify[Det[B2]] == 1, O22]

Out[11]= {{O22 → Csc[ph] (-1 + O21 Cos[ph] Csc[psi]) Sin[psi]}}

In[12]:= B3 = FullSimplify[B2 /. Flatten[erg2]]

Out[12]= {{Sin[ph]^2 + Cos[ph] (O21 Cos[psi] Cot[psi] - Cos[th] Cot[psi] Sin[ph] + O21 Sin[psi]),
     Cot[ph] (-1 + O21 Cos[ph] Csc[psi]) + Cos[ph] (-Cos[ph] Cos[th] Cot[psi] + Sin[ph]), 0},
    {Sin[ph] (Cos[ph] - O21 Cos[psi] Cot[psi] + Cos[th] Cot[psi] Sin[ph] - O21 Sin[psi]),
     1 + Cos[ph] (Cos[ph] - O21 Csc[psi] + Cos[th] Cot[psi] Sin[ph]), 0}, {0, 0, 1}}

In[13]:= erg3 = FullSimplify[Solve[B3[[1]][[1]] == B3[[2]][[2]], O21]]

Out[13]= {{O21 → Cos[psi] Cos[th] Sin[ph] + Cos[ph] Sin[psi]}}

In[14]:= B4 = FullSimplify[B3 /. Flatten[erg3]]

Out[14]= {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}

In[15]:= MatrixForm[FullSimplify[Omat /. Flatten[erg1] /. Flatten[erg2] /. Flatten[erg3]]]

Out[15]//MatrixForm=
( Cos[ph] Cos[psi] - Cos[th] Sin[ph] Sin[psi]   -Cos[psi] Sin[ph] - Cos[ph] Cos[th] Sin[psi]   Sin[psi] Sin[th]  )
( Cos[psi] Cos[th] Sin[ph] + Cos[ph] Sin[psi]   Cos[ph] Cos[psi] Cos[th] - Sin[ph] Sin[psi]    -Cos[psi] Sin[th] )
(              Sin[ph] Sin[th]                               Cos[ph] Sin[th]                       Cos[th]       )
```

In[16]:= **MatrixForm[Dmat]**

Out[16]//MatrixForm=

$$\begin{pmatrix} \text{Cos[ph] Cos[psi]} - \text{Cos[th] Sin[ph] Sin[psi]} & -\text{Cos[psi] Sin[ph]} - \text{Cos[ph] Cos[th] Sin[psi]} & \text{Sin[psi] Sin[th]} \\ \text{Cos[psi] Cos[th] Sin[ph]} + \text{Cos[ph] Sin[psi]} & \text{Cos[ph] Cos[psi] Cos[th]} - \text{Sin[ph] Sin[psi]} & -\text{Cos[psi] Sin[th]} \\ \text{Sin[ph] Sin[th]} & \text{Cos[ph] Sin[th]} & \text{Cos[th]} \end{pmatrix}$$

In[17]:= **DmatLock1 = FullSimplify[Dmat /. {th → 0}]**

Out[17]= {{Cos[ph + psi], -Sin[ph + psi], 0}, {Sin[ph + psi], Cos[ph + psi], 0}, {0, 0, 1}}

In[18]:= **DmatLock2 = FullSimplify[Dmat /. {th → Pi}]**

Out[18]= {{Cos[ph - psi], -Sin[ph - psi], 0}, {-Sin[ph - psi], -Cos[ph - psi], 0}, {0, 0, -1}}

In[19]:= **MatrixForm[DmatLock2]**

Out[19]//MatrixForm=

$$\begin{pmatrix} \text{Cos[ph - psi]} & -\text{Sin[ph - psi]} & 0 \\ -\text{Sin[ph - psi]} & -\text{Cos[ph - psi]} & 0 \\ 0 & 0 & -1 \end{pmatrix}$$